
F07 – Linear Equations (LAPACK)

F07UJF (STPTRI/DTPTRI) – NAG Fortran Library Routine Document

Note. Before using this routine, please read the Users’ Note for your implementation to check the interpretation of bold
italicised terms and other implementation-dependent details.

1 Purpose

F07UJF (STPTRI/DTPTRI) computes the inverse of a real triangular matrix, using packed storage.

2 Specification

SUBROUTINE F07UJF(UPLO, DIAG, N, AP, INFO)
ENTRY stptri(UPLO, DIAG, N, AP, INFO)
INTEGER N, INFO
real AP(∗)
CHARACTER∗1 UPLO, DIAG

The ENTRY statement enables the routine to be called by its LAPACK name.

3 Description

This routine forms the inverse of a real triangular matrix A using packed storage. Note that the inverse
of an upper (lower) triangular matrix is also upper (lower) triangular.

4 References

[1] Du Croz J J and Higham N J (1992) Stability of methods for matrix inversion IMA J. Numer.
Anal. 12 1–19

5 Parameters

1: UPLO — CHARACTER*1 Input

On entry: indicates whether A is upper or lower triangular as follows:

if UPLO = ’U’, then A is upper triangular;
if UPLO = ’L’, then A is lower triangular.

Constraint: UPLO = ’U’ or ’L’.

2: DIAG — CHARACTER*1 Input

On entry: indicates whether A is a non-unit or unit triangular matrix as follows:

if DIAG = ’N’, then A is a non-unit triangular matrix;
if DIAG = ’U’, then A is a unit triangular matrix; the diagonal elements are not referenced
and are assumed to be 1.

Constraint: DIAG = ’N’ or ’U’.

3: N — INTEGER Input

On entry: n, the order of the matrix A.

Constraint: N ≥ 0.

4: AP(∗) — real array Input/Output

Note: the dimension of the array AP must be at least max(1,N∗(N+1)/2).
On entry: the n by n triangular matrix A, packed by columns. More precisely, if UPLO = ’U’, the
upper triangle of A must be stored with element aij in AP(i+ j(j − 1)/2) for i ≤ j;if UPLO = ’L’,
the lower triangle of A must be stored with element aij in AP(i + (2n − j)(j − 1)/2) for i ≥ j. If
DIAG = ’U’, the diagonal elements of the matrix are not referenced and are assumed to be 1;the
same storage scheme is used whether DIAG = ’N’ or ’U’.

On exit: A is overwritten by A−1, using the same storage format as described above.

5: INFO — INTEGER Output

On exit: INFO = 0 unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

INFO < 0

If INFO = −i, the ith parameter had an illegal value. An explanatory message is output, and
execution of the program is terminated.

INFO > 0

If INFO = i, aii is zero and the matrix A is singular.

7 Accuracy

The computed inverse X satisfies
|XA − I| ≤ c(n)ε|X ||A|,

where c(n) is a modest linear function of n, and ε is the machine precision.

Note that a similar bound for |AX − I| cannot be guaranteed, although it is almost always satisfied.

The computed inverse satisfies the forward error bound

|X − A−1| ≤ c(n)ε|A−1||A||X |.

See Du Croz and Higham [1].

8 Further Comments

The total number of floating-point operations is approximately 1
3n3.

The complex analogue of this routine is F07UWF (CTPTRI/ZTPTRI).

9 Example

To compute the inverse of the matrix A, where

A =




4.30 0.00 0.00 0.00
−3.96 −4.87 0.00 0.00
0.40 0.31 −8.02 0.00

−0.27 0.07 −5.95 0.12


 ,

using packed storage.

9.1 Program Text

Note. The listing of the example program presented below uses bold italicised terms to denote precision-dependent details.
Please read the Users’ Note for your implementation to check the interpretation of these terms. As explained in the Essential
Introduction to this manual, the results produced may not be identical for all implementations.

* F07UJF Example Program Text
* Mark 15 Release. NAG Copyright 1991.
* .. Parameters ..

INTEGER NIN, NOUT
PARAMETER (NIN=5,NOUT=6)
INTEGER NMAX
PARAMETER (NMAX=8)
CHARACTER DIAG
PARAMETER (DIAG=’N’)

* .. Local Scalars ..
INTEGER I, IFAIL, INFO, J, N
CHARACTER UPLO

* .. Local Arrays ..
real AP(NMAX*(NMAX+1)/2)

* .. External Subroutines ..
EXTERNAL stptri, X04CCF

* .. Executable Statements ..
WRITE (NOUT,*) ’F07UJF Example Program Results’

* Skip heading in data file
READ (NIN,*)
READ (NIN,*) N
IF (N.LE.NMAX) THEN

*
* Read A from data file
*

READ (NIN,*) UPLO
IF (UPLO.EQ.’U’) THEN

READ (NIN,*) ((AP(I+J*(J-1)/2),J=I,N),I=1,N)
ELSE IF (UPLO.EQ.’L’) THEN

READ (NIN,*) ((AP(I+(2*N-J)*(J-1)/2),J=1,I),I=1,N)
END IF

*
* Compute inverse of A
*

CALL stptri(UPLO,DIAG,N,AP,INFO)
*
* Print inverse
*

WRITE (NOUT,*)
IFAIL = 0
CALL X04CCF(UPLO,DIAG,N,AP,’Inverse’,IFAIL)

END IF
STOP

*
END

9.2 Program Data

F07UJF Example Program Data
4 :Value of N
’L’ :Value of UPLO
4.30

-3.96 -4.87
0.40 0.31 -8.02

-0.27 0.07 -5.95 0.12 :End of matrix A

9.3 Program Results

F07UJF Example Program Results

Inverse
1 2 3 4

1 0.2326
2 -0.1891 -0.2053
3 0.0043 -0.0079 -0.1247
4 0.8463 -0.2738 -6.1825 8.3333

